

PROC FREQ erklärt (mit) Fußball

Manuela Wern	Stefanie Daum
SCHUFA Holding AG	SCHUFA Holding AG
Massenbergstr. 13a	Massenbergstr. 13a
44787 Bochum	44787 Bochum
manuela.wern@schufa.de	stefanie.daum@schufa.de

Zusammenfassung

Die Prozedur FREQ ist eine der bekanntesten SAS-Prozeduren. In diesem Beitrag geht es darum die Möglichkeiten von PROC FREQ darzustellen und damit Thesen aus der Welt des Fußballs zu überprüfen. Es werden verschiedene Formen der Häufigkeitsauszählung – sowohl tabellarisch als auch grafisch – und die Ausgabemöglichkeiten als SAS-Datei erklärt.

Schlüsselwörter: BASE SAS, PROC FREQ, ODS GRAPHICS

1 Einführung

Die Prozedur FREQ ist eine der bekanntesten SAS-Prozeduren. Jeder SAS-Anwender lernt schon früh einige Verwendungsmöglichkeiten dieser Prozedur kennen, aber werden auch wirklich alle Möglichkeiten ausgeschöpft?

In praktischen kleinen Beispielen sollen die vielfältigen Einsatzmöglichkeiten der Prozedur FREQ vorgestellt werden z.B.

- Häufigkeitsauszählungen mit verschiedenen Optionen
- Ausgabemöglichkeiten des Ergebnisses als SAS-Datei
- graphische Darstellung mit PROC FREQ und ODS
- Statistische Kennzahlen mit PROC FREQ

Die Prozedur wird folgendermaßen aufgerufen:

```
PROC FREQ <Optionen>;
  TABLES Kombinationen/Optionen;
  BY Variablen;
  WEIGHT Variable;
  FORMAT Variable Format;
RUN;
```

Verschiedene Anwendungsbeispiele dieser Prozedur werden mit Hilfe von Fußball-Daten vorgestellt. Folgende Daten standen für die Auswertungen zur Verfügung:

- Spielergebnisse aus den letzten 10 Saisons (1)
- Herbstmeister und Meister seit der Saison 1963/1964 (2)
- Meister, Vizemeister, Dritter und Vierter seit der Saison 1991/1992 (3)
- Aufsteiger in die 1. Bundesliga seit der Saison 1963/1964 (4)
- Endresultate der Ergebnisse der 1. Bundesliga seit der Saison 1963/1964 (5)

Quellen:

- (1) <http://www.football-data.co.uk> / <http://fussball-ergebnisse.t-online.de/>
- (2) <http://www.bundesliga.de/de/statistik/saison/index.php>
- (3) http://de.wikipedia.org/wiki/Liste_der_deutschen_Fu%C3%9Fballmeister
- (4) <http://www.trainer-baade.de/alle-aufsteiger-in-die-1-bundesliga/>
- (5) <http://www.mister-bulicheck.de>

2 Ist das häufigste Endergebnis nach 90 Minuten 2:1?

Bei Tippspielen wird häufig das Ergebnis 2:1 getippt, aber ist das auch das häufigste Ergebnis? Eine Auszählung mit Hilfe von PROC FREQ soll hier Antworten geben. Der Prozeduraufruf sieht folgendermaßen aus:

```
PROC FREQ DATA = Bundesliga;
    TABLES Ergebnis;
RUN;
```

Hinter der TABLES-Anweisung folgen die auszuzählenden Variablen. In diesem Fall sollen die Spielresultate der letzten zehn Bundesligasaisons ausgewertet werden. Das Ergebnis ist eine einfache Häufigkeitsauszählung.

Tabelle 1: Häufigkeitsauszählung der Spielresultate

Ergebnis	Frequency	Percent	Cumulative Frequency	Cumulative Percent
0:0	203	6.63	203	6.63
0:1	195	6.37	398	13.01
0:2	140	4.58	538	17.58
0:3	60	1.96	598	19.54
0:4	18	0.59	616	20.13
0:5	4	0.13	620	20.26
0:6	6	0.20	626	20.46
1:0	273	8.92	899	29.38
1:1	352	11.50	1251	40.88
1:2	197	6.44	1448	47.32
1:3	104	3.40	1552	50.72
:				
6:2	5	0.16	3050	99.67
6:3	3	0.10	3053	99.77
7:0	4	0.13	3057	99.90
7:1	1	0.03	3058	99.93
7:4	1	0.03	3059	99.97
8:1	1	0.03	3060	100.00

Da in dieser Tabelle das häufigste Ergebnis nicht sofort zu finden ist, ist eine Sortierung nach Häufigkeit sinnvoll. Das wird mit der Option ORDER = FREQ erreicht.

Der Aufruf sieht dann folgendermaßen aus:

```
PROC FREQ DATA = Bundesliga ORDER = FREQ;
    TABLES Ergebnis;
RUN;
```

Das Ergebnis der Auszählung erscheint nun nach (absteigenden) Häufigkeiten sortiert.

Tabelle 2: Häufigkeitsauszählung der Spielergebnisse geordnet nach Häufigkeiten

Ergebnis	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1:1	352	11.50	352	11.50
1:0	273	8.92	625	20.42
2:1	266	8.69	891	29.12
2:0	228	7.45	1119	36.57
0:0	203	6.63	1322	43.20
1:2	197	6.44	1519	49.64
0:1	195	6.37	1714	56.01
2:2	176	5.75	1890	61.76

Es ist zu sehen, dass das häufigste Ergebnis 1:1 war. Hierbei ist zu beachten, dass für diese Auswertung z. B. die Ergebnisse 1:0 und 0:1 getrennt ausgezählt werden.

Kleine Anmerkung: Zählt man diese als gleiches Ergebnis, so ist übrigens das häufigste Ergebnis der letzten zehn Saisons 1:0 bzw. 0:1, dicht gefolgt von 2:1 bzw. 1:2.

3 Werden Spiele schon in der ersten Halbzeit entschieden?

Eine weitere interessante Frage ist, ob Sieg, Niederlage oder Unentschieden schon nach der ersten Halbzeit feststehen.

```
PROC FREQ DATA = Bundesliga;
  TABLES Halbzeitergebnis * Endergebnis;
RUN;
```

Hier werden nach der TABLES-Anweisung zwei Variablen mit einem * verknüpft. Dies führt zu einer Kreuztabelle der beiden Merkmale.

Tabelle 3: Kreuztabelle der Halbzeit- und Endergebnisse

Table of Halbzeitergebnis by Endergebnis

Halbzeitergebnis		Endergebnis			
Frequency	Percent				
Row Pct					
Col Pct	A	U	H	Total	
A	499	153	71	723	
	16.31	5.00	2.32	23.63	
	69.02	21.16	9.82		
	56.83	19.90	5.02		
U	313	451	476	1240	
	10.23	14.74	15.56	40.52	
	25.24	36.37	38.39		
	35.65	58.65	33.69		

H	66	165	866	1097
	2.16	5.39	28.30	35.85
	6.02	15.04	78.94	
	7.52	21.46	61.29	
Total	878	769	1413	3060
	28.69	25.13	46.18	100.00

A = Auswärtssieg, H = Heimsieg, U = Unentschieden

In dieser Tabelle werden standardmäßig pro Zelle vier Zahlen dargestellt: die absolute Häufigkeit der einzelnen Zellen, der prozentuale Anteil der einzelnen Zelle an der Gesamtzahl der Beobachtungen, die Zeilen- und Spaltenprozentage. Dies führt in diesem Fall dazu, dass das Ergebnis etwas unübersichtlich ist.

In diesem Beispiel sind die Zeilen- und Spaltenprozentage nicht relevant, so dass die Ausgabe dieser Werte unterdrückt werden kann. Das geschieht mit den Optionen NOROW NOCOL hinter der TABLES-Anweisung.

```
PROC FREQ DATA = Bundesliga;
  TABLES Halbzeitergebnis * Endergebnis / NOROW NOCOL;
RUN;
```

Die aus diesem Aufruf resultierende Tabelle ist somit deutlich übersichtlicher.

Tabelle 4: Kreuztabelle der Halbzeit- und Endergebnisse ohne Zeilen- und Spaltenprozentage

Table of Halbzeitergebnis by Endergebnis

Halbzeitergebnis	Endergebnis			
Frequency	A	D	H	Total
Percent				
A	499	153	71	723
	16.31	5.00	2.32	23.63
D	313	451	476	1240
	10.23	14.74	15.56	40.52
H	66	165	866	1097
	2.16	5.39	28.30	35.85
Total	878	769	1413	3060
	28.69	25.13	46.18	100.00

Aus dieser Tabelle ist zu erkennen, dass immerhin bei 40,65% der Spiele die Tendenz sich in der zweiten Halbzeit noch ändert, bei 59,35% der Spiele (Summe der Prozente der Zellen auf der Diagonalen) bleibt die Tendenz gleich.

4 Aufsteiger – Kommt das Ende der Euphorie in der Rückrunde?

Die Euphorie nach dem Aufstieg in die erste Bundesliga ist groß und die erste Zeit oft auch sehr erfolgreich. Nur hält dies dauerhaft an? Um diese Frage zu beantworten, ist eine Kreuztabelle der Platzierung in der Hinrunde und der Platzierung in der Rückrunde hilfreich. Betrachtet werden die Aufsteiger seit der ersten Bundesligasaison 1963/1964.

```
PROC FREQ DATA = Aufsteiger;
  TABLES Platz_Hinrunde * Platz_Rueckrunde;
RUN;
```

Das Ergebnis hierzu ist allerdings sehr unübersichtlich und mehrere Seiten lang. Hier ein Auszug:

Tabelle 5: Kreuztabelle der Platzierung in der Hin- und Rückrunde

Table of Platz_Hinrunde by Platz_Rueckrunde

Platz_Hinrunde		Platz_Rueckrunde						
Frequency	Percent	1	3	4	5	6	7	Total
Row Pct	Col Pct							
1	1	1	0	0	0	0	1	2
	0.81	0.00	0.00	0.00	0.00	0.00	0.81	1.63
	50.00	0.00	0.00	0.00	0.00	0.00	50.00	
	100.00	0.00	0.00	0.00	0.00	0.00	33.33	
2	0	1	0	0	0	0	0	1
	0.00	0.81	0.00	0.00	0.00	0.00	0.00	0.81
	0.00	100.00	0.00	0.00	0.00	0.00	0.00	
	0.00	100.00	0.00	0.00	0.00	0.00	0.00	
3	0	0	1	0	0	0	0	2
	0.00	0.00	0.81	0.00	0.00	0.00	0.00	1.63
	0.00	0.00	50.00	0.00	0.00	0.00	0.00	
	0.00	0.00	50.00	0.00	0.00	0.00	0.00	
4	0	0	1	0	0	0	0	1
	0.00	0.00	0.81	0.00	0.00	0.00	0.00	0.81
	0.00	0.00	100.00	0.00	0.00	0.00	0.00	
	0.00	0.00	50.00	0.00	0.00	0.00	0.00	
5	0	0	0	1	0	0	0	1
	0.00	0.00	0.00	0.81	0.00	0.00	0.00	0.81
	0.00	0.00	0.00	100.00	0.00	0.00	0.00	
	0.00	0.00	0.00	33.33	0.00	0.00	0.00	
6	0	0	0	1	0	0	0	3
	0.00	0.00	0.00	0.81	0.00	0.00	0.00	2.44
	0.00	0.00	0.00	33.33	0.00	0.00	0.00	
	0.00	0.00	0.00	33.33	0.00	0.00	0.00	
Total	1	1	2	3	1	3		123
	0.81	0.81	1.63	2.44	0.81	2.44		100.00

(Continued)

Um das Ergebnis übersichtlicher zu machen, ist es zunächst hilfreich die Prozente, die Spalten- und die Zeilenprozente wegzulassen. Dies wird, wie schon zuvor, erreicht mit den Optionen NOROW, NOCOL und NOPERCENT in der TABLES-Anweisung. Das allein ist allerdings nicht ausreichend. Ein Format für die Platzierung macht das Ergebnis übersichtlicher. Das wird mit der FORMAT-Anweisung erreicht, wo für die beiden Variablen Platz_Hinrunde und Platz_Rueckrunde ein zuvor definiertes Format angegeben wird.

```
PROC FREQ DATA=Aufsteiger;
  TABLES Platz_Hinrunde * Platz_Rueckrunde
    / NOROW NOCOL NOPERCENT;
  FORMAT Platz_Hinrunde Platz_Rueckrunde platz.;
RUN;
```

Das neue Ergebnis lässt sich einfacher interpretieren.

Tabelle 6: Kreuztabelle der Platzierung in der Hin- und Rückrunde mit hinterlegtem Format

Table of Platz_Hinrunde by Platz_Rueckrunde

Platz_Hinrunde	Platz_Rueckrunde						
Frequency	1-3	4-6	7-9	10-12	13-15	16-20	Total
1-3	2	1	2	0	0	0	5
4-6	0	3	0	2	0	0	5
7-9	0	1	5	6	6	1	19
10-12	0	0	3	8	8	4	23
13-15	0	0	2	4	8	12	26
16-20	0	1	1	2	9	32	45
Total	2	6	13	22	31	49	123

Die Zahlen auf der Diagonale wurden nachträglich fett markiert, um die Interpretation zu erleichtern. Es ist zu erkennen, dass die Platzierung in der Rückrunde in den meisten Fällen schlechter als in der Hinrunde ist.

5 Wer Herbstmeister wird, gewinnt auch die Meisterschaft?

Ob der Herbstmeister in der Regel auch Meister wird, soll eine Betrachtung der Herbstmeister bzw. Meister der Bundesliga zeigen. Seit SAS 9.2 können mit Hilfe des Output Delivery System (ODS) und PROC FREQ direkt (statistische) Grafiken erzeugt werden.

Zunächst wird ein Balkendiagramm mit relativen Häufigkeiten erstellt, die die Meister der Bundesliga zeigt.

```

ODS GRAPHICS ON;
PROC FREQ DATA = Bundesliga;
  TABLES Meister / PLOTS(ONLY) = FREQPLOT (SCALE = PERCENT);
RUN;
ODS GRAPHICS OFF;

```

Neben der normalen Tabelle mit den Häufigkeiten, wird folgende Grafik erzeugt:




Abbildung 1: Häufigkeitsauszählung der Fußball-Bundesligameister

Diese Grafik hilft noch nicht bei der Beantwortung der oben gestellten Frage. Deshalb sollen nun die Variablen Meister und Herbst_Meister miteinander kombiniert werden. Die Variable Herbst_Meister hat zwei Ausprägungen (Ja und Nein). Diese geben an, ob die jeweilige Meistermannschaft auch Herbstmeister war. Eine Erweiterung des obigen Programmschritts um die Variable Herbst_Meister würde die gezeigte Grafik nun in zwei Grafiken teilen; für jede Ausprägung der Variablen Herbst_Meister eine eigene Grafik. Ist eine Grafik mit gestapelten Balken gewünscht, so kann das durch die Option `TWOWAY = STACKED` erreicht werden:

```

ODS GRAPHICS ON;
PROC FREQ DATA = Bundesliga;
  TABLES Herbst_Meister * Meister / PLOTS(ONLY) = FREQPLOT

```

```
(SCALE = PERCENT TWOWAY = STACKED);  
RUN;  
ODS GRAPHICS OFF;
```

Die erzeugte Grafik sieht so aus:




Abbildung 2: Kombinierte Häufigkeitsauszählung der Variablen Meister und Herbst_Meister

In rund einem Drittel der Fälle wird der Herbstmeister also doch nicht Meister, wobei unter Berücksichtigung des Punktevorsprung von Bayern München zu sagen bleibt, dass der Meister in diesem Jahr wohl schon feststeht.

6 Bekommt die Auswärtsmannschaft häufiger eine rote Karte?

Mit der Prozedur PROC FREQ lassen sich auch statistische Tests durchführen. So ein Test soll die Frage beantworten, ob der Heimvorteil sich auf die Anzahl der roten Karten erstreckt. Ausgewertet werden hierfür die letzten zehn Bundesligasaisons, wobei für zwei Jahre keine Daten über rote Karten vorliegen. Bei der Beantwortung dieser Frage kann der χ^2 -Test helfen. Getestet wird die Nullhypothese, dass die erwarteten Häufigkeiten der roten Karten für Auswärts- und Heimmannschaft gleich sind, also

$$H_0: P_{\text{Rot_Heim}} = P_{\text{Rot_Auswärts}}$$

Die Nullhypothese wird auf dem Niveau α (z. B. = 0,05) abgelehnt, falls die Überschreitungswahrscheinlichkeit „Prob“ $\leq \alpha$ ist, d.h. die Wahrscheinlichkeit für die beobachteten Daten unter der Nullhypothese klein ist.

Um den χ^2 -Test durchzuführen, muss in der TABLES-Anweisung die Option CHISQ angegeben werden.

```
PROC FREQ DATA = Karten;
  TABLES Karte_rot * Mannschaft / CHISQ;
RUN;
```

Dieser Programmschritt erzeugt neben der Kreuztabelle noch folgenden Output:

Tabelle 7: Ergebnisse der statistischen Tests

Statistics for Table of Karte_rot by Mannschaft

Statistic	DF	Value	Prob
Chi-Square	2	25.5286	<.0001
Likelihood Ratio Chi-Square	2	25.7424	<.0001
Mantel-Haenszel Chi-Square	1	25.2648	<.0001
Phi Coefficient		0.0681	
Contingency Coefficient		0.0679	
Cramer's V		0.0681	

Effective Sample Size = 5508
Frequency Missing = 612

Mit < 0.0001 ist die Überschreitungswahrscheinlichkeit kleiner als $\alpha = 0.05$, somit kann die Nullhypothese abgelehnt werden, woraus geschlossen wird, dass die Wahrscheinlichkeiten bzw. erwarteten Häufigkeiten der roten Karten für Heim- und Auswärtsmannschaft also nicht gleich sind.

Mit folgendem Aufruf lässt sich das Ergebnis noch einmal verdeutlichen

```
PROC FREQ DATA = Karten;
  TABLES Karte_rot*Mannschaft / EXPECTED DEVIATION CELLCHI2
                                NOCOL NOROW NOPERCENT;
RUN;
```

Mit den Optionen EXPECTED, DEVIATION und CELLCHI2 lässt sich die Kreuztabelle modifizieren, so dass in jeder Zelle neben den Häufigkeiten auch die erwarteten Häufigkeiten, die Abweichungen der tatsächlichen von der erwarteten Häufigkeiten sowie die Beiträge der Zellen zum Gesamt- χ^2 -Wert dargestellt werden.

Tabelle 8: Kreuztabelle der Variablen Karte_rot und Mannschaft mit optionalen Kennzahlen

Table of Karte_rot by Mannschaft

Karte_rot	Mannschaft		
	Auswärts	Heim	Total
Frequency			
Expected			
Deviation			
Cell Chi-Square			
-----	-----	-----	-----
0	2468	2572	5040
	2520	2520	
	-52	52	
	1.073	1.073	
-----	-----	-----	-----
1	272	175	447
	223.5	223.5	
	48.5	-48.5	
	10.525	10.525	
-----	-----	-----	-----
2	14	7	21
	10.5	10.5	
	3.5	-3.5	
	1.1667	1.1667	
-----	-----	-----	-----
Total	2754	2754	5508

Frequency Missing = 612

Bei den Spielen mit einer roten Karte ist die Abweichung von den erwarteten Häufigkeiten zu den tatsächlichen Häufigkeiten prozentual am größten. Diese Zellen leisten auch den größten Beitrag zur χ^2 -Teststatistik.

7 Die Offensive gewinnt Spiele und die Defensive gewinnt Meisterschaften

Was ist dran am viel zitierten Spruch: „Die Offensive gewinnt Spiele und die Defensive gewinnt Meisterschaften“? Hierfür wird eine Datei benötigt, in der für die letzten zehn Saisons alle Mannschaften mit der Anzahl der Tore und der Anzahl der Gegentore aufgeführt sind. Mit dem PROC FREQ lässt sich das Ergebnis einer Häufigkeitsauszählung in eine SAS-Datei umleiten. Dies geschieht mit der Anweisung OUT = gefolgt vom gewünschten Dateinamen im TABLES-Statement. Die in die neue SAS-Datei ausgegebenen Variablen lassen sich hierbei direkt umbenennen.

```
PROC FREQ DATA = Tore ORDER = FREQ;
  TABLES tor_durch / OUT = Tore
    (RENAME = (COUNT = Anzahl_Tore));
  TABLES tor_gegen / OUT = Gegentore
    (RENAME = (COUNT = Anzahl_Gegentore));
  BY saison;
RUN;
```

Die so erzeugten Dateien werden innerhalb eines Datenschritts mit einem MERGE-Statement zusammengeführt. Zusätzlich werden in dieser Datei die Platzierungen bzgl. Offensive/Defensive bestimmt und die tatsächliche Platzierung am Ende der Saison ermittelt. Mit dieser neu erzeugten Datei kann dann folgender Programmschritt durchgeführt werden:

```
PROC FREQ DATA = Tore_Gegentore;
TABLES offensiv defensiv;
WHERE Platz = 1;
RUN;
```

Mit der WHERE-Anweisung kann das Ergebnis in diesem Fall auf die erstplatzierte Mannschaft eingeschränkt werden, was zu folgenden Outputs führt:

Tabelle 9: Häufigkeitsauswertung der Platzierung nach Offensive und Defensive der Meister der letzten zehn Jahre

Offensiv	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	6	60.00	6	60.00
2	4	40.00	10	100.00

Defensiv	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	4	40.00	4	40.00
2	2	20.00	6	60.00
3	4	40.00	10	100.00

Die Betrachtung der letzten zehn Saisons hinterlässt den Eindruck, dass die Offensive wichtiger ist: sechsmal ist die Mannschaft mit der besten Offensive auch Meister geworden, viermal die Mannschaft mit der zweitbesten. Dagegen ist die Mannschaft mit der besten Defensive nur viermal Meister geworden, viermal war es sogar nur die Mannschaft mit der drittbesten Defensive. Demnach ist der oben zitierte Spruch für die letzten zehn Saisons nicht zutreffend.

Eine Erweiterung auf alle 49 Bundesligasaisons führt zu folgenden Outputs:

Tabelle 10: Häufigkeitsauswertung der Platzierung nach Offensive und Defensive der Meister der letzten 49 Jahre

Offensiv	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	23	46.94	23	46.94
2	14	28.57	37	75.51
3	6	12.24	43	87.76
4	4	8.16	47	95.92
7	1	2.04	48	97.96
8	1	2.04	49	100.00

Defensiv	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	28	57.14	28	57.14
2	8	16.33	36	73.47
3	9	18.37	45	91.84
4	3	6.12	48	97.96
7	1	2.04	49	100.00

Jetzt sieht das Bild schon etwas anders aus. In 28 Fällen hat die Mannschaft mit der besten Defensive auch die Meisterschaft gewonnen und nur 23-mal die Mannschaft mit der besten Offensive.

8 Meisterliches ...

In diesem Kapitel sollen noch weitere Möglichkeiten von PROC FREQ dargestellt werden.

Die im folgenden Prozedurschritt angegebenen Option PAGE bewirkt, dass das Ergebnis jeder der drei angegebenen Variablen auf einer neuen Seite ausgegeben wird. Ohne diese Anweisung würden die ersten zwei Auszählungen auf einer Seite stehen.

```
PROC FREQ DATA = Herbstmeister
      ORDER = FREQ PAGE;
      TABLES Meister Herbstmeister Trainer;
RUN;
```

Eine weitere Möglichkeit den FREQ Output direkt anzupassen ist die Option LIST. Folgender Programmschritt erzeugt eine ziemlich unübersichtliche Kreuztabelle:

```
PROC FREQ DATA = Herbstmeister ORDER = FREQ;
      TABLES Trainer * Meister;
      WHERE trainer in ( "Udo Lattek"
                        "Ottmar Hitzfeld"
                        "Hennes Weisweiler" );
RUN;
```

Tabelle 11: Kreuztabelle der Meistertrainer und der dazugehörigen Mannschaften

Table of Trainer by Meister

Trainer	Meister				Total
Frequency					
Percent					
Row Pct					
Col Pct	FC Bayer n Münche n	Borussia M'gladb ach	Borussia Dortmun d	1.FC Köl n	
Udo Lattek	6 31.58 75.00 54.55	2 10.53 25.00 40.00	0 0.00 0.00 0.00	0 0.00 0.00 0.00	8 42.11
Ottmar Hitzfeld	5 26.32 71.43 45.45	0 0.00 0.00 0.00	2 10.53 28.57 100.00	0 0.00 0.00 0.00	7 36.84
Hennes Weisweiler	0 0.00 0.00 0.00	3 15.79 75.00 60.00	0 0.00 0.00 0.00	1 5.26 25.00 100.00	4 21.05
Total	11 57.89	5 26.32	2 10.53	1 5.26	19 100.00

Mit der LIST-Option in der TABLES-Anweisung lässt sich dieses Ergebnis in Form einer Liste ausgeben.

```
PROC FREQ DATA = Herbstmeister ORDER = FREQ;
TABLES Trainer * Meister / LIST NOPERCENT NOCUM NOROW;
WHERE trainer IN ( "Udo Lattek"
                  "Ottmar Hitzfeld"
                  "Hennes Weisweiler" );
RUN;
```

Dieser Programmcode erzeugt folgende übersichtlichere Liste:

Tabelle 12: Auflistung der Meistertrainer und der dazugehörigen Mannschaften

The FREQ Procedure

Trainer	Meister	Frequency
Udo Lattek	FC Bayern München	6
Udo Lattek	Borussia M'gladbach	2
Ottmar Hitzfeld	FC Bayern München	5
Ottmar Hitzfeld	Borussia Dortmund	2
Hennes Weisweiler	Borussia M'gladbach	3
Hennes Weisweiler	1.FC Köln	1

Literatur

- [1] SAS Institute Inc.: SAS 9.3 Help and Documentation
- [2] J. Dufner, U. Jensen, E. Schumacher: Statistik mit SAS, Verlag Teubner, 2002
- [3] C. Ortseifen: Der SAS-Kurs – Eine leicht verständliche Einführung, International Thomson Publishing GmbH, 1997
- [4] A. Graf, W. Bundschuh, H.-G. Kruse: Effektives arbeiten mit SAS, B.I: Hochschultaschenbuch Band 646, BI Wissenschafts-Verlag, 1993